DECUS
PROGRAM LIBRARY

DECUS NO.	8-492
TITLE	BINLOAD, BINTAPE AND SEARCH
AUTHOR	William B. Wright Submitted by: John W. Taylor
COMPANY	Bucks County Community College Newtown, Pennsylvania
DATE	
SOURCE LANGUAGE	PAL III

DECUS Program Library Write-up DECUS NO. 8-492

BINLOAD, BINTAPE, AND SEARCH

Abstract

This program is designed to replace the binary loader, DEC-08-LBAA-PM for use in a PDP-8e with 4K memory. In addition, the program contains a core to paper-tape dump routine designed to replace DEC-08-YX1A-PB, which can be used to obtain a paper tape in standard binary format for any or all of the 4K core. Also contained is a routine for searching core for any desired 12-bit word. The address of each occurrence of the word is displayed in the accumulator. Entire routine resides in locations 7600-7777, hence is compatible with other DEC routines. A 12-instruction toggle-in bootstrap loader for the PDP-8e for loading this routine is also provided.

Directions for use

Toggle in manually the bootstrap loader, starting at address 0000. Place the binary tape for BINLOAD in the TTY tape reader. Using a starting address of 0000, start the machine. When the BINLOAD tape has run through, the entire routine has been loaded. Note that the actual length of this tape is critical due to limitations of the bootstrap loader, and depends upon the automatic switch on the TTY to stop loading when the tape runs out.

To load a binary tape using this routine, use a starting address of 7600. It is necessary to start your tape at the first actual address, however, which means bypassing the first address on binary paper tapes supplied by DEC, which is the address 0200 (see Figure 1 and Figure 2).* After the program is first started, the tape reader will run momentarily and then stop. This is a programmed halt at 7610 for the purpose of entering a 12-bit binary word which is to be added to each address of the tape being loaded, thus providing a relocatable feature. This feature is quite useful in the case where binary tapes of the various routines supplied by DEC under Digital-8-XX-U are kept on hand, all with a starting address of 0200. Any of these can be placed in core on any page without the necessity of assembling them each time. This feature must be used with caution, however, particularly if there are any absolute addresses in the user's program, and almost always must consist only of a change of page location because no provision is made for changing the target addresses in jump or JMS instructions. In the more usual case, no relocation is desired, so set the switch register to all down (0000) before pressing the continue key.

To use the BINTAPE portion of the program, a starting address of 7777 is used. This provides a jump to 7660, which is the actual start of the routine in the program. To use, set 7777 in the switch register. Then press load address key, clear key, and the continue key. The TTY will blip once. Turn on the

*The user will probably find it expedient to make new binary tapes, using the BINTAPE portion of this program, of the DEC programs which he uses regularly, such as PAL III, Editor, FOCAL, etc. Paper tapes made with this routine run directly on the BINLOAD portion with no necessity to be careful of the starting position.

TTY punch. Set the beginning address for the core to paper-tape dump in the switch register, then press continue. Nothing happens here! Set the last address for the core to paper-tape dump in the switch register. Press continue again. Binary tape will now be produced on the TTY punch. When the tape stops, the area of core, together with its beginning address as specified, will have been punched. If other segments of core are desired on the same binary tape, set the starting address of the next segment in the switch register and press continue again, then mount the next ending address in the switch register and press continue again. Be careful not to press the "clear" key when loading additional segments, as this would destroy the checksum. When all segments of core have been punched as desired, set the switch register to 0000 and once again press the continue key. The TTY will commence producing punches in the 8-column of the paper tape until it is stopped manually by pressing the halt key. Because of this last feature, a starting address of 0000 for any segment of a core dump is not allowed, but this should not be a problem since that address is reserved for the interrupt feature, and hence will probably not be used by the programmer.

To use the SEARCH routine, start by setting the switch register to 0012 and press load-address and examine keys. Note the contents of this address so that it can be restored when finished searching, then enter 0000 in 0012. Next enter the desired 12-bit word in address 7776. Using 7765 as a starting address, start the program. The address of the next occurrence of the 12-bit word entered in 7776 will be displayed in the AC register. Each time the continue key is pressed, the next occurrence of the word will be displayed until all 4K of memory has been examined, then it willstart over again. Restore the contents of 0012 and return to the program you were executing before the search.

Program Limitations

This program utilizes an instruction peculiar to the PDP-8e, namely the byte-swap (BSW) in two locations, hence cannot be used as written on any of the other PDP-8 machines without modification. RTL instructions can be substituted but then the SEARCH routine would not have sufficient room, and would have to be eliminated.

This program is not designed to handle more than 4K of core, and will not recognize the 7-8 bit combination which indicates other fields of core. Any 8-punch in a binary tape being read in will cause a halt, hence diagnostic messages from the Assembler will cause a halt. It is not difficult to circumvent this by simply lifting the binary tape out of the reader when it halts, and manually moving the tape to the first row of holes past the end of the diagnostic message (which will be indicated by a rub-out), and pressing the continue key.

Here is a listing of the bootstrap loader;

```
0000  6032          KCC             /USE THIS ROUTINE
0001  6031 START,   KSF             /TO TOGGLE-IN FOR
0002  5201          JMP  .-1        /LOADING BINLOAD
0003  6036          KRB             /& BINTAPE.  THIS
0004  7002          BSW             /PROVIDES NO AUTO-
0005  7421          MQL             /MATIC STOP.  TAPE
0006  6031          KSF             /MUST BE STOPPED
0007  5206          JMP  .-1        /MANUALLY OR BY
0010  6036          KRB             /MEANS OF THE AUTO-
0011  7501          MQA             /MATIC STOP ON THE
0012  3614          DCA I PNTR      /TTY.
0013  5201          JMP START
0014  7577 PNTR,    7577
```



Figure 1.

Start of typical paper tape as supplied by DEC or PAL III Assembler. To load by this program, start must be made at point indicated by arrow



Figure 2.

Start of a paper tape made by this program. Note elimination of starting 7-punch and the address 0200; also the absence of the 8-punches.

Both of the above examples are from the same program and have an actual starting address of 0001 as the point where loading of c ore is to commence.

```
                *12
0012  0000  PTR,      0000
                *7600
7600  7300            CLA CLL
7601  3360            DCA CSUM    /ZERO CHECKSUM STORAGE.
7602  6032            KCC         /CLEAR KEYBOARD BUFFER AND ACC.
7603  4350  LOAD,     JMS READ    /READ TTY BUFFER AND
7604  7450            SNA         /IF IT'S ZERO GO BACK TO LOAD.
7605  5203            JMP LOAD
7606  7402            HLT         /STOP FOR ENTERING RELOC-
7607  5211            JMP HERE    /ATABLE FACTOR. SEE LINE 7627.
7610  4350  HIGH,     JMS READ    /READ TTY BUFFER.
7611  3357  HERE,     DCA LOC     /AND STORE IN LOC.
7612  1357            TAD LOC
7613  0364            AND K300    /MASK FOR 7 OR 8 BIT.
7614  7450            SNA         /IF NO 7 OR 8 BIT THEN
7615  5240            JMP LODE    /ITS TO BE LOADED,
7616  0362            AND K100    /ELSE IS IT 7-BIT?
7617  7450            SNA         /IF NOT, ITS AN 8-BIT,
7620  5233            JMP EOJ     /HENCE WE'RE THROUGH.
7621  7300  ADDR,     CLA CLL     /WE'RE HERE IF IT'S A
7622  1357            TAD LOC     /7-BIT,HENCE THIS IS AN
7623  0361            AND K77     /ADDRESS WHICH IS TO BE
7624  3357            DCA LOC     /STORED IN PNTR.
7625  4244            JMS SMBL    /SO ASSEMBLE THE 12 BITS.
7626  3356            DCA PNTR    /ENTER THE RELOCATABLE FACTOR
7627  7404            OSR         /TO BE ADDED TO THE STARTING ADD
7630  1356            TAD PNTR    /RESSES ON THE TAPE, AND THEN
7631  3356            DCA PNTR    /STORE IN PNTR.
7632  5210            JMP HIGH    /GO BACK FOR NEXT CHARACTER.
7633  1360  EOJ,      TAD CSUM    /WE'RE HERE IF THERE WAS AN
7634  7110            CLL RAR     /8-BIT, WITH TAPE CHECKSUM IN AC.
7635  7041            CIA         /HENCE COMPLEMENT AND ADD OUR
7636  1342            TAD PNCH    /CHECKSUM FOR A ZERO AC, THEN
7637  7402            HLT         /STOP.
7640  4244  LODE,     JMS SMBL    /THIS ROUTINE LOADS ASSEMBLED
7641  3756            DCA I PNTR  /CHARACTER IN LOCATION
7642  2356            ISZ PNTR    /POINTED TO BY PNTR AND
7643  5210            JMP HIGH    /THEN INCREMENTS PNTR.
7644  0000  SMBL,     0000
7645  1357            TAD LOC     /GET 6-BIT CHARACTER
7646  7002            BSW         /AND SHIFT IT TO THE HIGH ORDER 6-BITS
7647  3357            DCA LOC     /FOR COMPLETE WORD, THEN STORE
7650  4350            JMS READ    /TEMPORARILY, READ LOW 6-BITS
7651  1357            TAD LOC     /THEN ADD TO HIGH ORDER 6-BITS.
7652  3342            DCA PNCH    /NOW STORE ASSEMBLED 12-BIT WORD.
7653  1342            TAD PNCH    /ADD IT TO FORMER VALUE OF CHECK-
7654  1360            TAD CSUM    /SUM AND PUT BACK IN
7655  3360            DCA CSUM    /CHECKSUM, THEN
7656  1342            TAD PNCH    /LEAVE SUBROUTINE WITH ASSEMBLED
7657  5644            JMP I SMBL  /12-BIT WORD IN ACCUMULATOR.
7660  6046  STRT,     TLS         /SET TTY PUNCH FLAG.
```

```
7661  7300          CLA CLL
7662  3360          DCA CSUM    /CLEAR CHECKSUM STORAGE.
7663  7402  RNTR,   HLT         /WAIT FOR STARTING ADDRESS TO BE
7664  7404          OSR         /ENTERED IN SWITCH REGISTER.
7665  7450          SNA         /CHECK FOR LAST ENTRY, IF LAST
7666  5317          JMP END     /GO TO END ROUTINE.
7667  3356          DCA PNTR    /IF NOT ZERO, SAVE IT IN PNTR
7670  7402          HLT         /WAIT FOR ADDRESS OF LAST WORD TO
7671  7404          OSR         /BE ENTERED IN SWITCH REGISTER.
7672  7040          CMA         /ESTABLISH COUNTER BY SUBTRACTING
7673  1356          TAD PNTR    /LAST ADDRESS FROM FIRST
7674  3355          DCA CNTR    /AND STORE IN COUNTER.
7675  1356          TAD PNTR    /ADD VALUE OF ADDRESS TO VALUE
7676  1360          TAD CSUM    /OF CHECKSUM AND DEPOSIT
7677  3360          DCA CSUM    /RESULT IN CHECKSUM.
7700  1356          TAD PNTR    /NOW PUT POINTER IN TEMPORARY
7701  3357          DCA LOC     /STORAGE FOR USE BY SET ROUTINE.
7702  7120          STL         /THIS MARKS IT AS AN ADDRESS
7703  4327          JMS SET     /TO PUNCH 7-BIT.
7704  1756  TAPE,   TAD I PNTR  /GET NEXT CORE LOCATION
7705  1360          TAD CSUM    /AND ADD IT TO CHECKSUM
7706  3360          DCA CSUM    /THEN PUT CHECKSUM BACK
7707  7100          CLL         /TO AVOID CONFUSION WITH 7-BIT
7710  1756          TAD I PNTR  /GET CORE LOCATION BACK AND
7711  3357          DCA LOC     /PUT IT IN LOC,
7712  4327          JMS SET     /THEN GO PUNCH IT.
7713  2356          ISZ PNTR    /INCREMENT THE POINTER AND THE
7714  2355          ISZ CNTR    /COUNTER UNTIL COUNTER REACHES 0,
7715  5304          JMP TAPE    /BUT GO BACK FOR NEXT IF NOT 0
7716  5263          JMP RNTR    /OR GO BACK FOR NEXT ADDRESS.
7717  1360  END,    TAD CSUM    /YOU'RE FINISHED, SO GET
7720  3357          DCA LOC     /CHECKSUM AND GO
7721  4327          JMS SET     /PUNCH IT.
7722  1363          TAD K200    /THEN SET UP FOR PUNCHING THE
7723  6041          TSF         /8-BIT TRAILER UNTIL MACHINE
7724  5323          JMP .-1     /IS STOPPED MANUALLY.
7725  6046          TLS
7726  5323          JMP .-3
7727  0000  SET,    0000
7730  1357          TAD LOC     /BRING IN NUMBER TO ROTATE
7731  7002          BSW         /HIGH ORDER BITS TO RIGHT SIDE,
7732  0361          AND K77     /MASK FOR RIGHT SIDE
7733  7430          SZL         /IF LINK IS SET THIS IS AN
7734  1362          TAD K100    /ADDRESS & NEEDS 7-BIT.
7735  4342          JMS PNCH    /PUNCH HIGH ORDER 6-BITS.
7736  1357          TAD LOC     /BRING IN NUMBER AGAIN
7737  0361          AND K77     /AND MASK LOW ORDER 6-BITS,
7740  4342          JMS PNCH    /THEN PUNCH THEM,
7741  5727          JMP I SET   /AND RETURN.
7742  0000  PNCH,   0000        /SUBROUTINE TO PUNCH A
7743  6041          TSF         /CHARACTER ON THE TTY AND
7744  5343          JMP .-1     /RETURN TO MAIN PROGRAM.
```

```
7745  6046            TLS
7746  7300            CLA CLL
7747  5742            JMP I PNCH
7750  0000  READ,     0000        /SUBROUTINE TO READ A CHARACTER
7751  6031            KSF         /FROM THE TELETYPE.
7752  5351            JMP .-1
7753  6036            KRB
7754  5750            JMP I READ /AND RETURN TO PROGRAM.
7755  0000  CNTR,     0000
7756  0000  PNTR,     0000
7757  0000  LOC,      0000
7760  0000  CSUM,     0000
7761  0077  K77,      0077
7762  0100  K100,     0100
7763  0200  K200,     0200
7764  0300  K300,     0300
7765  7300  SRCH,     CLA CLL     /THIS IS THE START OF A SEARCH
7766  1412            TAD I PTR   /ROUTINE TO BE ADDED TO THE END OF
7767  7041            CIA         /BINLOAD & BINTAPE. ENTER SEARCHED-
7770  1376            TAD DSRD    /FOR CHARACTER IN 7776. POINTER FOR
7771  7440            SZA         /ROUTINE IS IN 0012-THIS SHOULD BE
7772  5365            JMP SRCH    /CLEARED BEFORE STARTING. STARTING
7773  1012            TAD PTR     /ADDRESS IS 775.POUINE STOPS AND
7774  7402            HLT         /DISPLAYS ADDRESS OF SEARCHED-FOR
7775  5365            JMP SRCH    /CHARACTER IN AC. PRESS CONTINUE FOR
7776  0000  DSRD,     0000        /REPEATED OCCURENCES.
7777  5260            JMP STRT    /JUMPS TO START OF CORE PUNCH ROUTINE.
```

```
ADDR    7621
CNTR    7755
CSUM    7760
DSRD    7776
END     7717
EOJ     7633
HERE    7611
HIGH    7610
K100    7762
K200    7763
K300    7764
K77     7761
LOAD    7603
LOC     7757
LODE    7640
PNCH    7742
PNTR    7756
PTR     0012
READ    7750
RNTR    7663
SET     7727
SMBL    7644
SRCH    7765
STRT    7660
TAPE    7704
```

Note: The list of constants in addresses 7755-7764 are shared by both the BINLOAD and BINTAPE programs.